Práctica 6

Cálculos de límites y de integrales

6.1. Enunciado de la práctica

En esta práctica se trata de obtener con ordenador lo siguiente:

1. El cálculo de los límites de funciones:

   (a) $\lim\limits_{x\to 0} \dfrac{1}{x}$ (b) $\lim\limits_{x\to 0^+} \dfrac{1}{x}$

   (c) $\lim\limits_{x\to 0^-} \dfrac{1}{x}$ (d) $\lim\limits_{x\to 0^+} \dfrac{1}{\operatorname{sen}^2 x} - \dfrac{1}{x^2}$

   (e) $\lim\limits_{x\to 0} \cos(2x)^{3/x^2}$ (f) $\lim\limits_{x\to 0} x^{\operatorname{sen} x}$

   (g) $\lim\limits_{x\to 3} \dfrac{\sqrt{x}-\sqrt{3}-\sqrt{x-3}}{\sqrt{x^2-9}}$ (h) $\lim\limits_{x\to a} \dfrac{a^x - a^{\operatorname{sen} x}}{x^3}$

2. El cálculo de los límites de sucesiones:

   (a) $\lim\limits_{n\to +\infty} \dfrac{\sqrt[n]{n!}}{n}$ (b) $\lim\limits_{n\to +\infty} \left(\dfrac{1}{n+1} + \dfrac{1}{n+2} + \cdots + \dfrac{1}{2n}\right)$

   (c) $\lim\limits_{n\to +\infty} n(\ln(n+3) - \ln(n-1))$ (d) $\lim\limits_{n\to +\infty} \left(\dfrac{\sqrt[n]{2}+\sqrt[n]{3}+\sqrt[n]{4}}{3}\right)^n$

   (e) $\lim\limits_{n\to +\infty} \sqrt[n]{\dfrac{n^2+n+1}{n+1}}$

3. El cálculo de las primitivas:

   (a) $\displaystyle\int \frac{\ln x}{x^2}\,dx$ (b) $\displaystyle\int \frac{\operatorname{sen}(2x)}{1+\operatorname{sen}^2 x}\,dx$

   (c) $\displaystyle\int \frac{2x^3-7x^2+9x+1}{(x^2+x+x)(x-2)^2}\,dx$ (d) $\displaystyle\int \sqrt{x^2+x+1}\,dx$

4. El cálculo del volumen del sólido situado en el semiespacio $z \geq 0$ y limitado por los cilindros $C_1\colon x^2+y^2=1$, $C_2\colon (x-1)^2+(y-1)^2=1$ y por la superficie $z = xy$.

5. El cálculo de la longitud de la curva plana de ecuaciones paramétricas:

$$\left\{\begin{array}{lcl} x(t) & = & \cos t + \dfrac{1}{2}\cos 7t + \dfrac{1}{3}\operatorname{sen} 17t \\ & & \\ y(t) & = & \operatorname{sen} t + \dfrac{1}{2}\operatorname{sen} 7t + \dfrac{1}{3}\cos 17t \end{array}\right. \qquad t \in [0, 2\pi]$$

   Este problema puede verse enunciado en [EP2], p. 660.

6. El cálculo de la cantidad de metano que hay en la atmósfera de un nuevo planeta descubierto en una galaxia lejana y del que se sabe:

   a) Su forma es la de una esfera sólida de 3000 ⟦km⟧ de radio.

   b) La profundidad de la atmósfera es de 30 ⟦km⟧.

   c) La densidad de metano, en gramos por metro cúbico, a una altura sobre la superficie de h metros se estima en
   $$\delta = \frac{1.023}{h^2+1}$$

   En [Gr] p. 227 y 283 puede encontrarse una versión más general de este enunciado y en `http://www-math.bgsu.edu/~jgresse/MapleExample12(13).pdf` una resolución al cálculo del volumen de la atmósfera.

7. El cálculo del campo gravitatorio que una esfera hueca S, homogénea de masa M y de radio R induce sobre una partícula K situada en un punto que dista r del centro de la esfera.

6.2. Cálculo de límites de funciones

```
> hh:=Limit(sin(x),x=infinity): hh=value(hh);
```

$$\lim_{x\to\infty} \sin(x) = -1..1$$

El comando **limit** calcula el límite, **Limit** escribe la expresión literal, **value** realiza el cálculo exacto de expresiones literales mientras que **evalf** realiza el cálculo aproximado.

```
> hh:=Limit(1/x,x=0): hh=value(hh);
```

$$\lim_{x\to 0} \frac{1}{x} = \mathit{undefined}$$

```
> hh:=Limit(1/x,x=0,right): hh=value(hh);
```

$$\lim_{x\to 0+} \frac{1}{x} = \infty$$

En MAPLE, ∞ significa $+\infty$

```
> hh:=Limit(1/x,x=0,left): hh=value(hh);
```

$$\lim_{x\to 0-} \frac{1}{x} = -\infty$$

```
> f:=(1/sin(x)^2)-(1/x^2):
> hh:=Limit(f,x=0,right): hh=value(hh);
```

$$\lim_{x\to 0+} \frac{1}{\sin(x)^2} - \frac{1}{x^2} = \frac{1}{3}$$

```
> f:=(cos(2*x))^(3/x^2):
> hh:=Limit(f,x=0): hh=value(hh);
```

$$\lim_{x\to 0} \cos(2\,x)^{(\frac{3}{x^2})} = e^{(-6)}$$

```
> f:=x^sin(x):
> hh:=Limit(f,x=0):hh=value(hh);
```

$$\lim_{x\to 0} x^{\sin(x)} = 1$$

```
> f:=(sqrt(x)-sqrt(3)-sqrt(x-3))/sqrt(x^2-9):
> hh:=Limit(f,x=3): hh=value(hh);
```

$$\lim_{x\to 3} \frac{\sqrt{x}-\sqrt{3}-\sqrt{x-3}}{\sqrt{x^2-9}} = -\frac{\sqrt{6}}{6}$$

```
> f:=(a^x-a^(sin(x)))/x^3: hh:=Limit(f,x=0):
> hh=value(hh);
```

$$\lim_{x\to 0} \frac{a^x - a^{\sin(x)}}{x^3} = \frac{1}{6}\ln(a)$$

6.3. Cálculo de límites de sucesiones

```
> assume(n>0); assume(n,integer);
```

Con **assume** suponemos que n es un número natural. Según podemos observar, MAPLE marca las variables con condiciones con el símbolo ˜.

```
> f:= (n!)^(1/n)/n:
> hh:=Limit(f,n=+infinity): hh=value(hh);
```

$$\lim_{n\tilde{}\to\infty} \frac{(n\tilde{}!)^{(\frac{1}{n\tilde{}})}}{n\tilde{}} = e^{(-1)}$$

```
> f:=sum(1/sqrt(n+k), k=1..n):
> hh:=Limit(f,n=+infinity): hh=value(hh);
```

$$\lim_{n\tilde{}\to\infty} \sum_{k=1}^{n\tilde{}} \frac{1}{\sqrt{n\tilde{}+k}} = \infty$$

```
> f:= n*(ln(n+3)-ln(n-1)):
> hh:=Limit(f,n=+infinity): hh=value(hh);
```

$$\lim_{n\tilde{}\to\infty} n\tilde{}\,(\ln(n\tilde{}+3)-\ln(n\tilde{}-1)) = 4$$

```
> f:=(1/3*(2^(1/n)+3^(1/n)+4^(1/n)))^n:
> hh:=Limit(f,n=+infinity): hh=value(hh);
```

$$\lim_{n\tilde{}\to\infty} \left(\frac{2^{(\frac{1}{n\tilde{}})}}{3} + \frac{3^{(\frac{1}{n\tilde{}})}}{3} + \frac{4^{(\frac{1}{n\tilde{}})}}{3}\right)^{n\tilde{}} = 2\,3^{(1/3)}$$

```
> f:=((n^2+n+1)/(n+1))^(1/n):
> hh:=Limit(f,n=+infinity): hh=value(hh);
```

$$\lim_{\tilde{n}\to\infty} (\frac{\tilde{n}^2 + \tilde{n} + 1}{\tilde{n} + 1})^{(\frac{1}{\tilde{n}})} = 1$$

6.4. Cálculo de primitivas

```
> f:= ln(x) / x^2:
> hh:=Int(f,x); hh=value(hh);
```

$$hh := \int \frac{\ln(x)}{x^2}\,dx$$

$$\int \frac{\ln(x)}{x^2}\,dx = -\frac{\ln(x)}{x} - \frac{1}{x}$$

```
> int(f,x);
```

$$-\frac{\ln(x)}{x} - \frac{1}{x}$$

Con el comando **int** se evalua la primitiva, con **Int** tan sólo se expresa

```
> f:=sin(2*x)/(1+sin(x)^2): hh:=Int(f,x):
> hh=value(hh);
```

$$\int \frac{\sin(2\,x)}{1+\sin(x)^2}\,dx = \ln(-2+\cos(x)^2)$$

```
> f:=(2*x^3-7*x^2+9*x+1)/((x^2+x+1)*(x-2)^2):
> hh:=Int(f,x): hh=value(hh);
```

$$\int \frac{2\,x^3 - 7\,x^2 + 9\,x + 1}{(x^2+x+1)\,(x-2)^2}\,dx = \ln(x^2+x+1) - \frac{2}{3}\sqrt{3}\arctan(\frac{(2\,x+1)\sqrt{3}}{3}) - \frac{1}{x-2}$$

```
> f:=sqrt(x^2+x+1):
> hh:=Int(f,x): hh=value(hh);
```

$$\int \sqrt{x^2+x+1}\,dx = \frac{(2\,x+1)\sqrt{x^2+x+1}}{4} + \frac{3}{8}\operatorname{arcsinh}\left(\frac{2\sqrt{3}\,(x+\frac{1}{2})}{3}\right)$$

6.5. Cálculo de integrales definidas

6.5.1. Resolución del punto 4. Cálculo del volumen

Nuestro sólido es un cuerpo situado sobre el plano XY cuya altura en cada punto (x, y) viene dado por la función $z(x, y) = xy$ y cuya base es la región R del plano XY limitada por las circunferencias unidad centradas en el origen, C_1: $x^2 + y^2 = 1$, y en el punto $(1, 1)$, C_2: $(x-1)^2 + (y-1)^2 = 1$, obteniendo

$$\mathrm{vol}(H) = \iint_R xy\,dx\,dy = \int_0^1 dx \int_{y_2}^{y_1} xy\,dy$$

donde $y_2 = 1 - \sqrt{1-(x-1)^2}$ es la circunferencia C_2 y $y_1 = \sqrt{1-x^2}$ es la circunferencia C_1. Desarrollando la cuenta anterior con MAPLE obtenemos:

```
> y[2]:=1-sqrt(1-(x-1)^2); y[1]:=sqrt(1-x^2);
> # y[2] es el límite inferior e y[1] el superior
```

$$y_2 := 1 - \sqrt{-x^2 + 2\,x}$$
$$y_1 := \sqrt{1 - x^2}$$

```
> hh:=Int(Int(x*y,y=y[2]..y[1]),x=0..1):
> Volumen:=value(hh);
> hh=evalf(%);
```

$$Volumen := \frac{\pi}{4} - \frac{2}{3}$$
$$\int_0^1 \int_{1-\sqrt{-x^2+2\,x}}^{\sqrt{1-x^2}} x\,y\,dy\,dx = 0.1187314968$$

6.5.2. Resolución del punto 5. Cálculo de la longitud de la curva elegante

```
> x:=cos(t)+(1/2)*cos(7*t)+(1/3)*sin(17*t);
> y:=sin(t)+(1/2)*sin(7*t)+(1/3)*cos(17*t);
> plot([x,y,t=0..2*Pi]);
```

$$x := \cos(t) + \frac{1}{2}\cos(7\,t) + \frac{1}{3}\sin(17\,t)$$
$$y := \sin(t) + \frac{1}{2}\sin(7\,t) + \frac{1}{3}\cos(17\,t)$$




```
> x2:=(diff(x,t))^2; y2:=(diff(y,t))^2;
> long:=Int(sqrt(x2+y2),t=0..2*Pi);
> #long=value(long);
```

$$x2 := (-\sin(t) - \frac{7}{2}\sin(7\,t) + \frac{17}{3}\cos(17\,t))^2$$
$$y2 := (\cos(t) + \frac{7}{2}\cos(7\,t) - \frac{17}{3}\sin(17\,t))^2$$
$$long := \int_0^{2\pi} \sqrt{(-\sin(t) - \frac{7}{2}\sin(7\,t) + \frac{17}{3}\cos(17\,t))^2 + (\cos(t) + \frac{7}{2}\cos(7\,t) - \frac{17}{3}\sin(17\,t))^2}\,dt$$

Si ejecutamos la línea `>long=value(long);` comprobamos que MAPLE nos repite el último resultado. Eso significa que MAPLE no es capaz de evaluar la integral de forma exacta. Para obtenerla de forma aproximada utilizamos el comando **evalf**

```
> long:=evalf(long);
```

$$long := 39.40357871$$

¿Cómo hace el programa para evaluar aproximadamente una integral definida? Utiliza una mejora del método de Simpson. Realizar una busqueda de Internet poniendo 'metodo simpson' con el fin de encontrar referencias sobre el tema.

6.5.3. Resolución del punto 6. Cantidad de metano en la atmósfera de un nuevo planeta

Para H la región del espacio limitada por la atmósfera, se tiene, utilizando coordenadas esféricas centradas en el centro del planeta:

$$\begin{aligned}
\text{Metano}\,[\![g]\!] &= \iiint_H \delta(h)\,dxdydz = \iiint_{H_{\rho,\phi,L}} \rho^2 \operatorname{sen}\phi \cdot \delta(h)\,d\rho d\phi dL \\
&\overset{1)}{=} 1.023 \iiint_{H_{h,\phi,L}} \operatorname{sen}\phi \frac{(\rho_0+h)^2}{h^2+1}\,dh d\phi dL \\
&= 1.023 \int_{-\pi}^{\pi} dL \int_0^{\pi} \operatorname{sen}\,\phi\,d\phi \int_0^{30000} \frac{(3000000+h)^2}{h^2+1}\,dh\,[\![g]\!] \\
&= 1.023 \cdot 2\pi \cdot 2 \cdot \int_0^{30000} \frac{(3000000+h)^2}{h^2+1}\,dh\,[\![g]\!] \\
&\overset{2)}{=} 0.1817358344\ 10^{15}\ [\![g]\!]
\end{aligned} \tag{6.1}$$

En 1) hacemos el cambio de variables $\rho = \rho_0 + h$ para $\rho_0 = \rho_0(\ell, L)$ el valor de ρ a altura cero, es decir sobre la superficie del planeta, en el caso de este problema $\rho_0(\ell, L) = 3000000\ [\![\mathrm{m}]\!]$. En 2) ponemos el resultado obtenido con las instrucciones:

```
> delta:=1.023/(h^2+1);
> rho:=3*10^6+h;
```

$$\delta := \frac{1.023}{h^2+1}$$

$$\rho := 3000000 + h$$

```
> hh:=Int(Int(Int(rho^2*delta*sin(phi),h=0..30*10^3),
> phi=0..Pi),L=-Pi..Pi):
> CantidadMetano:=hh=value(hh)*g;
```

$$CantidadMetano := \int_{-\pi}^{\pi}\int_0^{\pi}\int_0^{30000} \frac{1.023\,(3000000+h)^2 \sin(\phi)}{h^2+1}\,dh\,d\phi\,dL = 0.1817358344\,10^{15}\,g$$

6.5.4. Resolución del punto 7. Campo gravitatorio inducido por una esfera hueca

En la figura 6.1 hemos representado la fuerza gravitatoria ejercida sobre la partícula K por el elemento dS de masa $\delta = \dfrac{M}{4\pi R^2}$ y situado en el punto P. Tomando un sistema de referencia cartesiano (X, Y, Z) con la condición que la semirrecta $\overrightarrow{OK}$ sea el semieje positivo



Figura 6.1: Atracción gravitatoria sobre K ejercida por dS.

de las zetas, podemos concretar el valor de la aceleración gravitatoria $g = g_x \frac{\partial}{\partial x} + g_y \frac{\partial}{\partial y} + g_z \frac{\partial}{\partial z}$ que la esfera S induce sobre K de la siguiente manera:

$$g = \begin{cases} -\dfrac{GM}{r^2}\dfrac{\partial}{\partial z} & \text{si } r \geq R \\ 0 & \text{si } r < R \end{cases} \tag{6.2}$$

donde G es la constante de gravitación universal.

En efecto, se verifica:

1) $g_x = g_y = 0$. Si denotamos por P^{YZ} al simétrico respecto al plano YZ del punto P y por $dS(P^{YZ})$ al elemento de superficie de S situado en P^{YZ}; entonces, la componente, según la dirección de X, de atracción gravitatoria debida al elemento $dS(P^{YZ})$ es opuesta a la debida a $dS(P)$. Ambas componentes se cancelan entre sí y, en consecuencia, $g_x = 0$. Análogamente, $g_y = 0$.

2) Se tiene:

$$g_z = \iint_S g_z(dS)\, dS = \int_{-\pi}^{\pi} dL \int_0^{\pi} g_z(dS)\, R^2 \operatorname{sen}\phi\, d\phi \tag{6.3}$$

para $g_z(dS)$ la componente, según Z, de la aceleración gravitatoria debida al elemento dS. La segunda de las igualdades en la relación anterior resulta de (**??**).

3) Claramente, la proyección de $g(dS)$ sobre el eje Z no depende de la longitud L en la que está situada dS, sólo depende del ángulo ϕ formado entre Z y $g(dS)$; en consecuencia, podemos suponer que dS está situado en un punto P del meridiano cero, es decir, en el plano XZ e identificando el punto (x, z) del plano XZ con el complejo $w = x + i\,z$ podemos poner:

$$K = i\,r; \quad P = Re^{i\,(\pi/2-\phi)} = i\,Re^{-i\,\phi}; \quad g_z(dS) = \frac{G\delta\,\mathrm{Im}(P-K)}{|P-K|^3} = \frac{G\delta\,\mathrm{Re}(e^{-i\,\phi} - \frac{r}{R})}{R^2\,|e^{-i\phi} - \frac{r}{R}|^3}$$

para Re, Im los operadores *parte real*, *parte imaginaria*, respectivamente. En consecuencia, para $c = r/R \geq 0$ tenemos:

$$
\begin{aligned}
g_z &= 2\pi \cdot G \cdot \delta \cdot \int_0^\pi \frac{\operatorname{Re}(e^{-i\phi}-c)}{|e^{-i\phi}-c|^3} \operatorname{sen}\phi \, d\phi \\
&= 2\pi G\delta \int_0^\pi \frac{(\cos\phi - c)\operatorname{sen}\phi}{[\operatorname{sen}^2\phi + (\cos\phi - c)^2]^{3/2}} \, d\phi \\
&\overset{1)}{=} 2\pi G\delta \int_{-1}^1 \frac{u-c}{(c^2+1-2cu)^{3/2}} \, du \\
&\overset{2)}{=} \frac{2\pi G\delta}{2c^2} \int_{\pm(c-1)}^{1+c} \frac{1-c^2-v^2}{v^2} \, dv \\
&= -\frac{GM}{r^2} \text{ (si } c>1 \text{ ó) } 0 \text{ (si } c<1)
\end{aligned}
\tag{6.4}
$$

lo que nos permite concluir la fórmula (6.2). En 1) hacemos el cambio de variables $u = \cos\phi$ y en 2) el cambio $v = \sqrt{c^2+1-2cu}$ con lo que el límite inferior vale $\sqrt{c^2-2c+1} = \pm(c-1)$ donde hemos de elegir el signo positivo si $c > 1$ y el negativo si $c < 1$.

Realizamos con Maple las cuentas anteriores:

En el exterior y sobre la esfera

```
> macro(pp=phi);
```

$$pp$$

```
> t:= exp(-I*pp)-c; rt:= Re(t);
```

$$t := e^{(-I\,\phi)} - c$$
$$rt := \Re(e^{(-I\,\phi)} - c)$$

```
> assume(c>=1);
> hh:= Int(rt*sin(pp)/abs(t)^3, pp=0..Pi):
> Ec:=hh= value(hh);
> AceleraciónExterior:=
> 2*Pi*G*M/(4*Pi*R^2)*subs(c=r/R, rhs(Ec));
```

$$Ec := \int_0^\pi \frac{(-c\tilde{} + \Re(e^{(-I\,\phi)}))\sin(\phi)}{|-e^{(-I\,\phi)} + c\tilde{}|^3} \, d\phi = -\frac{2}{c\tilde{}^2}$$

$$Aceleraci\acute{o}nExterior := -\frac{G\,M}{r^2}$$

Con el comando **subs** realizamos substituciones.

En el interior de la esfera

```
> assume(c<1,c>=0);
> hh:= Int(rt*sin(pp)/abs(t)^3, pp=0..Pi):
> Ec:=hh= value(hh);
> AceleraciónInterior:=
> 2*Pi*G*M/(4*Pi*R^2)*subs(c=r/R, rhs(Ec));
```

$$Ec := \int_0^\pi \frac{(-c\tilde{} + \Re(e^{(-I\,\phi)}))\sin(\phi)}{|e^{(-I\,\phi)} - c\tilde{}|^3} \, d\phi = 0$$

$$Aceleraci\acute{o}nInterior := 0$$